

Buckling

Introduction

This tutorial was created using ANSYS 7.0 to solve a simple buckling problem.

It is recommended that you complete the [NonLinear Tutorial](#) prior to beginning this tutorial


Buckling loads are critical loads where certain types of structures become unstable. Each load has an associated buckled mode shape; this is the shape that the structure assumes in a buckled condition. There are two primary means to perform a buckling analysis:

1. Eigenvalue

Eigenvalue buckling analysis predicts the theoretical buckling strength of an ideal elastic structure. It computes the structural eigenvalues for the given system loading and constraints. This is known as classical Euler buckling analysis. Buckling loads for several configurations are readily available from tabulated solutions. However, in real-life, structural imperfections and nonlinearities prevent most real-world structures from reaching their eigenvalue predicted buckling strength; ie. it over-predicts the expected buckling loads. This method is not recommended for accurate, real-world buckling prediction analysis.

2. Nonlinear

Nonlinear buckling analysis is more accurate than eigenvalue analysis because it employs non-linear, large-deflection, static analysis to predict buckling loads. Its mode of operation is very simple: it gradually increases the applied load until a load level is found whereby the structure becomes unstable (ie. suddenly a very small increase in the load will cause very large deflections). The true non-linear nature of this analysis thus permits the modeling of geometric imperfections, load perturbations, material nonlinearities and gaps. For this type of analysis, note that small off-axis loads are necessary to initiate the desired buckling mode.



This tutorial will use a steel beam with a 10 mm X 10 mm cross section, rigidly constrained at the bottom. The required load to cause buckling, applied at the top-center of the beam, will be calculated.

Eigenvalue Buckling Analysis

Preprocessing: Defining the Problem

1. Open preprocessor menu

`/PREP7`

2. Give example a Title

Utility Menu > File > Change Title ...

`/title,Eigen-Value Buckling Analysis`

3. Define Keypoints

Preprocessor > Modeling > Create > Keypoints > In Active CS ...

`K, #, X, Y`

We are going to define 2 Keypoints for this beam as given in the following table:

Keypoints	Coordinates (x,y)
1	(0,0)
2	(0,100)

4. Create Lines

Preprocessor > Modeling > Create > Lines > Lines > In Active Coord
L, 1, 2

Create a line joining Keypoints 1 and 2

5. Define the Type of Element

Preprocessor > Element Type > Add/Edit/Delete...

For this problem we will use the BEAM3 (Beam 2D elastic) element. This element has 3 degrees of freedom (translation along the X and Y axes, and rotation about the Z axis).

6. Define Real Constants

Preprocessor > Real Constants... > Add...

In the 'Real Constants for BEAM3' window, enter the following geometric properties:

- i. Cross-sectional area AREA: 100
- ii. Area moment of inertia IZZ: 833.333
- iii. Total Beam Height HEIGHT: 10

This defines a beam with a height of 10 mm and a width of 10 mm.

7. Define Element Material Properties

Preprocessor > Material Props > Material Models > Structural > Linear > Elastic > Isotropic

In the window that appears, enter the following geometric properties for steel:

- i. Young's modulus EX: 200000
- ii. Poisson's Ratio PRXY: 0.3

8. Define Mesh Size

Preprocessor > Meshing > Size Cntrl > ManualSize > Lines > All Lines...

For this example we will specify an element edge length of 10 mm (10 element divisions along the line).

9. Mesh the frame

Preprocessor > Meshing > Mesh > Lines > click 'Pick All'

LMESH, ALL

Solution Phase: Assigning Loads and Solving

1. Define Analysis Type


Solution > Analysis Type > New Analysis > Static

ANTYPE, 0

2. Activate prestress effects

To perform an eigenvalue buckling analysis, prestress effects must be activated.

- You must first ensure that you are looking at the unabridged **solution menu** so that you can select **Analysis Options** in the **Analysis Type** submenu. The last option in the solution menu will either be 'Unabridged menu' (which means you are currently looking at the abridged version) or 'Abridged Menu' (which means you are looking at the unabridged menu). If you are looking at the abridged menu, select the unabridged version.
- Select **Solution > Analysis Type > Analysis Options**
- In the following window, change the [SSTIF][PSTRES] item to 'Prestress ON', which ensures the stress stiffness matrix is calculated. This is required in eigenvalue buckling analysis.



3. Apply Constraints

Solution > Define Loads > Apply > Structural > Displacement > On Keypoints
Fix Keypoint 1 (ie all DOF constrained).


4. Apply Loads

Solution > Define Loads > Apply > Structural > Force/Moment > On Keypoints

The eigenvalue solver uses a unit force to determine the necessary buckling load. Applying a load other than 1 will scale the answer by a factor of the load.

Apply a vertical (FY) point load of -1 N to the top of the beam (keypoint 2).

The applied loads and constraints should now appear as shown in the figure below.



5. Solve the System

Solution > Solve > Current LS
SOLVE

6. Exit the Solution processor

Close the solution menu and click **FINISH** at the bottom of the Main Menu.
FINISH


Normally at this point you enter the postprocessing phase. However, with a buckling analysis you must re-enter the solution phase and specify the buckling analysis. Be sure to close the solution menu and re-enter it or the buckling analysis may not function properly.

7. Define Analysis Type

Solution > Analysis Type > New Analysis > Eigen Buckling
ANTYPE, 1

8. Specify Buckling Analysis Options

- Select **Solution > Analysis Type > Analysis Options**
- Complete the window which appears, as shown below. Select 'Block Lanczos' as an extraction method and extract 1 mode. The 'Block Lanczos' method is used for large symmetric eigenvalue problems and uses the sparse matrix solver. The 'Subspace' method could also be used, however it tends to converge slower as it is a more robust solver. In more complex analyses the Block Lanczos method may not be adequate and the Subspace method would have to be used.



9. Solve the System

Solution > Solve > Current LS
SOLVE


10. Exit the Solution processor

Close the solution menu and click **FINISH** at the bottom of the Main Menu.
FINISH

Again it is necessary to exit and re-enter the solution phase. This time, however, is for an expansion pass. An expansion pass is necessary if you want to review the buckled mode shape(s).

11. Expand the solution

- Select **Solution > Analysis Type > Expansion Pass...** and ensure that it is on. You may have to select the 'Unabridged Menu' again to make this option visible.
- Select **Solution > Load Step Opts > ExpansionPass > Single Expand > Expand Modes ...**
- Complete the following window as shown to expand the first mode



12. Solve the System

Solution > Solve > Current LS

SOLVE

Postprocessing: Viewing the Results

1. View the Buckling Load


To display the minimum load required to buckle the beam select **General Postproc > List Results > Detailed Summary**. The value listed under 'TIME/FREQ' is the load (41,123), which is in Newtons for this example. If more than one mode was selected in the steps above, the corresponding loads would be listed here as well.

/POST1

SET,LIST

2. Display the Mode Shape

- Select **General Postproc > Read Results > Last Set** to bring up the data for the last mode calculated.
- Select **General Postproc > Plot Results > Deformed Shape**



Non-Linear Buckling Analysis

Ensure that you have completed the [NonLinear Tutorial](#) prior to beginning this portion of the tutorial

Preprocessing: Defining the Problem

1. Open preprocessor menu

/PREP7

2. Give example a Title

Utility Menu > File > Change Title ...

/TITLE, Nonlinear Buckling Analysis

3. Create Keypoints

Preprocessor > Modeling > Create > Keypoints > In Active CS
K, #, X, Y

We are going to define 2 keypoints (the beam vertices) for this structure to create a beam with a length of 100 millimeters:

Keypoint	Coordinates (x,y)
1	(0,0)
2	(0,100)

4. Define Lines

Preprocessor > Modeling > Create > Lines > Lines > Straight Line
Create a line between Keypoint 1 and Keypoint 2.
L, 1, 2

5. Define Element Types

Preprocessor > Element Type > Add/Edit/Delete...

For this problem we will use the BEAM3 (Beam 2D elastic) element. This element has 3 degrees of freedom (translation along the X and Y axis's, and rotation about the Z axis). With only 3 degrees of freedom, the BEAM3 element can only be used in 2D analysis.

6. Define Real Constants

Preprocessor > Real Constants... > Add...

In the 'Real Constants for BEAM3' window, enter the following geometric properties:

- i. Cross-sectional area AREA: 100
- ii. Area Moment of Inertia IZZ: 833.333
- iii. Total beam height HEIGHT: 10

This defines an element with a solid rectangular cross section 10 x 10 millimeters.

7. Define Element Material Properties

Preprocessor > Material Props > Material Models > Structural > Linear > Elastic > Isotropic
In the window that appears, enter the following geometric properties for steel:

- i. Young's modulus EX: 200e3
- ii. Poisson's Ratio PRXY: 0.3

8. Define Mesh Size

Preprocessor > Meshing > Size Cntrls > Lines > All Lines...

For this example we will specify an element edge length of 1 mm (100 element divisions along the line).

ESIZE, 1

9. Mesh the frame

Preprocessor > Meshing > Mesh > Lines > click 'Pick All'
LMESH, ALL

Solution: Assigning Loads and Solving


1. Define Analysis Type

Solution > New Analysis > Static
ANTYPE, 0

2. Set Solution Controls

- o Select **Solution > Analysis Type > Sol'n Control...**

The following image will appear:




Ensure the following selections are made under the 'Basic' tab (as shown above)

- A. Ensure Large Static Displacements are permitted (this will include the effects of large deflection in the results)
- B. Ensure Automatic time stepping is on. Automatic time stepping allows ANSYS to determine appropriate sizes to break the load steps into. Decreasing the step size usually ensures better accuracy, however, this takes time. The Automatic Time Step feature will determine an appropriate balance. This feature also activates the ANSYS bisection feature which will allow recovery if convergence fails.
- C. Enter 20 as the number of substeps. This will set the initial substep to $1/20^{\text{th}}$ of the total load.
- D. Enter a maximum number of substeps of 1000. This stops the program if the solution does not converge after 1000 steps.
- E. Enter a minimum number of substeps of 1.
- F. Ensure all solution items are written to a results file.

Ensure the following selection is made under the 'Nonlinear' tab (as shown below)

- G. Ensure Line Search is 'On'. This option is used to help the Newton-Raphson solver converge.
- H. Ensure Maximum Number of Iterations is set to 1000

**NOTE**

There are several options which have not been changed from their default values. For more information about these commands, type `help` followed by the command into the command line.

3. Apply Constraints

Solution > Define Loads > Apply > Structural > Displacement > On Keypoints
Fix Keypoint 1 (ie all DOFs constrained).

4. Apply Loads

Solution > Define Loads > Apply > Structural > Force/Moment > On Keypoints

Place a -50,000 N load in the FY direction on the top of the beam (Keypoint 2). Also apply a -250 N load in the FX direction on Keypoint 2. This horizontal load will persuade the beam to buckle at the minimum buckling load.


The model should now look like the window shown below.

**5. Solve the System**

Solution > Solve > Current LS

SOLVE

The following will appear on your screen for NonLinear Analyses




This shows the convergence of the solution.


General Postprocessing: Viewing the Results

1. View the deformed shape


- To view the element in 2D rather than a line: **Utility Menu > PlotCtrls > Style > Size and Shape** and turn 'Display of element' ON (as shown below).



- General Postproc > Plot Results > Deformed Shape... > Def + undeformed
PLDISP, 1



- **View the deflection contour plot**
General Postproc > Plot Results > Contour Plot > Nodal Solu... > DOF solution, UY
PLNSOL, U, Y, 0, 1




Other results can be obtained as shown in previous linear static analyses.

Time History Postprocessing: Viewing the Results


As shown, you can obtain the results (such as deflection, stress and bending moment diagrams) the same way you did in previous examples using the General Postprocessor. However, you may wish to view time history results such as the deflection of the object over time.

1. Define Variables


- o Select: **Main Menu > TimeHist Postpro.** The following window should open automatically.




If it does not open automatically, select **Main Menu > TimeHist Postpro > Variable Viewer**

- o Click the add button  in the upper left corner of the window to add a variable.


- Double-click **Nodal Solution > DOF Solution > Y-Component of displacement** (as shown below) and click OK. Pick the uppermost node on the beam and click OK in the 'Node for Data' window.




- To add another variable, click the add button again. This time select **Reaction Forces > Structural Forces > Y-Component of Force**. Pick the lowermost node on the beam and click OK.
- On the Time History Variable window, click the circle in the 'X-Axis' column for FY_3. This will make the reaction force the x-variable. The Time History Variables window should now look like this:



2. Graph Results over Time

- Click on UY_2 in the Time History Variables window.
- Click the graphing button  in the Time History Variables window.
- The labels on the plot are not updated by ANSYS, so you must change them manually. Select **Utility Menu > PlotCtrls > Style > Graphs > Modify Axes** and re-label the X and Y-axis appropriately.



The plot shows how the beam became unstable and buckled with a load of approximately 40,000 N, the point where a large deflection occurred due to a small increase in force. This is slightly less than the eigen-value solution of 41,123 N, which was expected due to non-linear geometry issues discussed above.

Command File Mode of Solution

The above example was solved using a mixture of the Graphical User Interface (or GUI) and the command language interface of ANSYS. This problem has also been solved using the ANSYS command language interface that you may want to browse. Copy and paste following code into Notepad or a similar text editor and save it to your computer. Now go to '**File > Read input from...**' and select the file.

```
FINISH          ! These two commands clear current data
/CLEAR

/TITLE,Eigenvalue Buckling Analysis

/PREP7         ! Enter the preprocessor

ET,1,BEAM3     ! Define the element of the beam to be buckled
R,1,100,833.333,10 ! Real Consts: type 1, area (mm^2), I (mm^4), height (mm)
MP,EX,1,200000 ! Young's modulus (in MPa)
MP,PRXY,1,0.3  ! Poisson's ratio

K,1,0,0        ! Define the geometry of beam (100 mm high)
K,2,0,100

L,1,2          ! Draw the line

ESIZE,10      ! Set element size to 1 mm
LMESH,ALL,ALL ! Mesh the line

FINISH
/SOLU         ! Enter the solution mode

ANTYPE,STATIC ! Before you can do a buckling analysis, ANSYS
              ! needs the info from a static analysis
```

```

PSTRES,ON          ! Prestress can be accounted for - required
                   ! during buckling analysis
DK,1,ALL           ! Constrain the bottom of beam

FK,2,FY,-1         ! Load the top vertically with a unit load.
                   ! This is done so the eigenvalue calculated
                   ! will be the actual buckling load, since
                   ! all loads are scaled during the analysis.

SOLVE
FINISH

/SOLU              ! Enter the solution mode again to solve buckling
ANTYPE,BUCKLE      ! Buckling analysis
BUCOPT,LANB,1     ! Buckling options - subspace, one mode
SOLVE
FINISH

/SOLU              ! Re-enter solution mode to expand info - necessary
EXPASS,ON          ! An expansion pass will be performed
MXPAND,1          ! Specifies the number of modes to expand
SOLVE
FINISH

/POST1            ! Enter post-processor
SET,LIST           ! List eigenvalue solution - Time/Freq listing is the
                   ! force required for buckling (in N for this case).
SET,LAST           ! Read in data for the desired mode
PLDISP            ! Plots the deflected shape
NonLinear Buckling
FINISH            ! These two commands clear current data
/CLEAR

/TITLE, Nonlinear Buckling Analysis
/PREP7            ! Enter the preprocessor
ET,1,BEAM3        ! Define element as beam3

MP,EX,1,200000    ! Young's modulus (in Pa)
MP,PRXY,1,0.3     ! Poisson's ratio

R,1,100,833.333,10 ! area, I, height

K,1,0,0,0         ! Lower node
K,2,0,100,0       ! Upper node (100 mm high)

L,1,2             ! Draws line

ESIZE,1           ! Sets element size to 1 mm
LMESH,ALL         ! Mesh line

FINISH
/SOLU

ANTYPE,STATIC     ! Static analysis (not buckling)
NLGEOM,ON         ! Non-linear geometry solution supported
OUTRES,ALL,ALL    ! Stores bunches of output
NSUBST,20         ! Load broken into 5 load steps
NEQIT,1000        ! Use 20 load steps to find solution
AUTOTS,ON         ! Auto time stepping
LNSRCH,ON

۱۴ /ESHAPE,1       ! Plots the beam as a volume rather than line

```

```
DK,1,ALL,0           ! Constrain bottom

FK,2,FY,-50000       ! Apply load slightly greater than predicted
                     ! required buckling load to upper node
FK,2,FX,-250         ! Add a horizontal load (0.5% FY) to initiate
                     ! buckling

SOLVE
FINISH

/POST26              ! Time history post processor
RFORCE,2,1,F,Y       ! Reads force data in variable 2
NSOL,3,2,U,Y         ! Reads y-deflection data into var 3
XVAR,2               ! Make variable 2 the x-axis
PLVAR,3              ! Plots variable 3 on y-axis

/AXLAB,Y,DEFLECTION  ! Changes y label
/AXLAB,X,LOAD         ! Changes X label
/REPLOTT
```

Reference

<http://www.mece.ualberta.ca/tutorials/ansys>